Bringing HANK to Dynare

Normann Rion   Sébastien Villemot

Dynare

18 September 2024

Current status of heterogeneity in Dynare

- Historically, Dynare used for representative-agent models (incl. RANK)
- Dynare also handles finite heterogeneity
  - both Ricardian and hand-to-mouth consumers (TANK)
  - multiple production sectors
  - multiple countries
- Dynare can already compute approximated solutions to some models with rich heterogeneity (incl. HANK)
  - by truncating idiosyncratic histories, as proposed by Le Grand and Ragot (JEDC, 2022)
  - by approximating the distribution with a parametric family, as proposed by Winberry (Quant. Econ., 2018)
- However, today, most HANK models are solved outside Dynare, using various specialized toolkits

Objectives

Our goal is to bring HANK to Dynare, in line with the philosophy of the project, which means:

1. It should be possible to describe HANK models in the most natural way for an economist, building on the existing Dynare language
2. There should be an easy access to solution methods at the frontier of the accuracy/cost trade-off, without the user having to write numeric code

The Krussel and Smith (1998) model

- Infinitely-many heterogeneous consumers with idiosyncratic labor market risk, and incomplete insurance markets
- Aggregate risk: total factor productivity shock
- Compared to the original paper:
  - the idiosyncratic shock is a *continuous* labor productivity shock, instead of a discrete unemployment risk
  - the idiosyncratic and aggregate shocks are uncorrelated

Individual household problem

$$\max_{\{c_{i,t},k_{i,t}\}_{t\geq 0}} \mathbb{E}_0 \sum_{t\geq 0} \beta^t u(c_{i,t})$$

s.t.

$$u(c_{i,t}) = c_{i,t}^{1-\sigma}$$
$$k_{i,t} \geq 0$$
$$c_{i,t} + k_{i,t} = R_t k_{i,t-1} + W_t \exp(e_{i,t})$$
$$e_{i,t} = \rho^e e_{i,t-1} + \varepsilon^e_{i,t}$$

where $e_{i,t}$ is the (log of) individual labor efficiency

First order conditions:

$$\lambda_{i,t} = R_t c_{i,t}^{-\sigma}$$
$$\beta \mathbb{E}_t \lambda_{i,t+1} + \xi_{i,t} = \frac{\lambda_{i,t}}{R_t}$$
$$\xi_{i,t} = 0 \perp k_{i,t} > 0$$

where $\lambda_{i,t}$ (resp. $\xi_{i,t}$) is Lagrange multiplier on budget constraint (resp. liquidity constraint)

Aggregate equations

$$Y_t = Z_t K_{t-1}^{\alpha}$$
$$R_t = 1 + \alpha Z_t K_{t-1}^{\alpha-1} - \delta$$
$$W_t = (1-\alpha) Z_t K_{t-1}^{\alpha}$$
$$Y_t = C_t + K_t - (1-\delta) K_{t-1}$$
$$K_t = \int k_{i,t} \mathrm{d}i$$
$$\log Z_t = \rho^Z \log Z_{t-1} + \varepsilon_t^Z$$

Dynare code

Declarations and calibration

```
heterogeneity_dimension households;

var(heterogeneity=households) c k lambda xi e;
varexo(heterogeneity=households) eps_e;

var R W C Y K Z;
varexo eps_Z;

parameters alpha beta delta sigma rho_e sig_e rho_Z sig_Z;

alpha = 0.36; beta = 0.9644; delta = 0.0233;
sigma = 5;     rho_e = 0.966; sig_e = 0.503*sqrt(1-rho_e^2);
rho_Z = 0.8;  sig_Z = 0.014;
```

Dynare code

Heterogeneous equations and shocks

```
model(heterogeneity=households);
  c = (lambda/R)^(-1/sigma);
  beta*lambda(+1) + xi = lambda/R;
  xi*k = 0; // In a future version: xi = 0 ⊥ k > 0
  R*k(-1) + W*exp(e) = (lambda/R)^(-1/sigma) + k;
  e = rho_e*e(-1) + eps_e;
end;

shocks(heterogeneity=households);
  var eps_e; stderr sig_e;
end;
```

Dynare code

Aggregate equations and shocks

```
model;
  Y = Z*K(-1)^alpha;
  R = 1 + alpha*Z*K(-1)^(alpha-1) - delta;
  W = (1-alpha)*Z*K(-1)^alpha;
  Y = C + K - (1-delta)*K(-1);
  K = SUM(k);
  log(Z) = rho_Z*log(Z(-1)) + eps_Z;
end;

shocks;
  var eps_Z; stderr sig_Z;
end;
```

HANK solution methods

Most solution methods proceed in two steps:

1. Compute the steady state, at which:
   - aggregate variables do not change
   - *the distribution* of heterogeneous variables does not change
   - in the absence of aggregate shocks…
   - …but taking into account idiosyncratic shocks

   It is an infinite-dimensional object:
   - a scalar value for aggregate variables
   - a distribution for heterogeneous variables

   Typically computed via a global solution method
2. Compute the dynamics in reaction to an aggregate shock
   Typically computed via a perturbation approach
   $\Rightarrow$ valid for a small aggregate shock in the neighborhood of the steady state

What's already implemented

- For now, steady state has to be computed outside Dynare
- Method chosen for dynamics: Bhandari, Bourany, Evans et Golosov (NBER, 2023)
- Generalization of the RANK perturbation approach to the HANK case
  $\Rightarrow$ very natural fit for Dynare
- Rewritten in MATLAB/Octave (authors' original implementation in Julia)

Next steps

- Generalize Bhandari et al. (2023) to a larger class of models
- More flexibility in specification of heterogeneous equations
- Implement global method(s) for steady state resolution
- Implement other solution methods for model dynamics